# Maintainer: Lukas Fleischer <lfleischer@archlinux.org>
# Contributor: Tomislav Ivek <tomislav.ivek@gmail.com>

pkgname=python-fasteners
pkgver=0.19
pkgrel=1
pkgdesc="A python package that provides useful locks."
arch=('any')
url="https://github.com/harlowja/fasteners"
license=('Apache')
depends=('python-six')
makedepends=('python-build' 'python-installer' 'python-setuptools' 'python-wheel')
source=("$pkgname-$pkgver.tar.gz::https://github.com/harlowja/fasteners/archive/$pkgver.tar.gz")
sha512sums=('47db216ff5bad5c669ca2d9be0ee09feba4a3d55ed0eac8d3623bd3d112b0b87fb9019f8785d035ec32ad9fa9e4ae7d1c92a576c97eec29acc82711d7e71a465')

build() {
  cd "$srcdir/fasteners-$pkgver"
  python3 -m build --wheel --no-isolation
}

package() {
  cd "$srcdir/fasteners-$pkgver"
  python3 -m installer --destdir="$pkgdir" dist/*.whl
}
